
Introduction
A major design goal of Ada 95 (the revision to Ada) was to avoid or at least minimise the need for modifying the existing base of Ada 83 software to make it compatible with Ada 95.  This involved not only pursuing upward compatibility but also preserving implementation-dependent behaviour that can currently be relied upon.  This goal has largely been achieved; the revision has introduced 43 known incompatibilities of which only six are likely to occur in a normal Ada program.  Total upward compatibility would not have allowed the correction of certain errors and certainly would not have allowed the enhancements needed to satisfy many of the revision requirements.
The great majority of Ada 83 programs will not be significantly affected by these changes - the most likely incompatibilities being automatically detected at compile-time.  Moreover, tools are being developed to aid in the reliable detection of any problems and thereby smooth the process of transition.
This guide is aimed at projects currently writing or intending to write Ada applications where enhancement or maintenance is required beyond 1997.  Its purpose is to alert such projects to any upward incompatibilities between Ada 83 and Ada 95 so that Ada 83 software can be written avoiding these incompatibilities long before transitioning to Ada 95 is required.  A set of guidelines is included for avoiding all but the obscure incompatibilities.
The guide describes each of the upward incompatibilities in terms of the language change, a rationale for the change, a description and example of the incompatibility, an indication of how it can be avoided in existing Ada 83 programs, and the possibility of its automatic detection and removal.  Many of the proposed incompatibilities will not occur in normal programs - the Ada 83 semantics being known only to the most erudite of Ada programmers, so each incompatibility is classified according to how likely a normal Ada 83 program of a reasonable size would contain an occurrence of the incompatibility.  Incompatibilities are also classified according to whether they cause illegalities at compile-time or different semantics at run-time.
This version of the guide corresponds to the International Standard version of the language, published in early 1995 [2].  The opportunity has been taken to present the incompatibilities in the order of the appropriate section of the Ada 83 Language Reference Manual [1].
References
[1]	ANSI/MIL-STD-1815A, Reference Manual for the Ada Programming Language, Department of Defense, January 1983
[2]	Programming Language Ada, ISO/IEC 8652-1995.

Summary of Upward Incompatibilities
A change (to a given feature) is upward compatible if the required behaviour of all legal Ada 83 programs is unaffected by the change.  In particular, all legal Ada 83 programs remain legal, with the same meaning.  Otherwise the change is upward incompatible.  
An upward incompatible change is further classified as being:
	upward consistent if the semantics of all legal Ada 83 programs that remain legal are unaffected by the change.  Some legal Ada 83 programs may become illegal, but those that remain legal have the same meaning.
	upward consistent if no exceptions if the semantics of all legal Ada 83 programs that do not raise a exception relating to the feature and that remain legal are unaffected by the change.  Some legal Ada 83 programs may become illegal, and some that raised an exception may no longer do so if a meaningful result can be produced.
	upward inconsistent if the semantics of some legal, correct Ada 83 programs are affected by the change.  In particular, the program may now raise a predefined exception when it didn't in the past, or it may produce a different result. 
This section gives a brief description of the upward incompatible changes from Ada 83 to Ada 95 grouped according to the above classification, together with a list of any incompatibilities that cannot be avoided in Ada 83 programs or that cannot be precisely detected.
Upward Inconsistent Incompatibilities
If a legal Ada 83 program contains occurrences of any of the upward inconsistent incompatibilities, its run-time behaviour will change as an Ada 95 program.  An incompatibility, for which most occurrences would be detected at compile time but where it is theoretically possible (but very unlikely) that it would cause a change in run time behaviour, is listed under Upward Consistent Incompatibilities and marked with a diamond ().
Derived type inherits all operations of parent - in Ada 95 a derived type inherits all its parent's primitive operations previously declared in the same declarative part, unlike Ada 83 (see Incompatibility 4).  It is unlikely that many programs derive from a type in the same package in which it is declared.
Real types may have less precision - the chosen representation of a real type may have less precision in Ada 95 compared to that in Ada 83 (see Incompatibilities 9 and 11).  This change will have little impact.
Default Small for fixed point subtypes - in Ada 83, the default value of Small for a fixed point subtype S is defined to be the largest power of two not exceeding S'Delta.  In Ada 95, it can be a smaller power of two (see Incompatibility 12).  Other than affecting the value of S'Small, this change will have little impact.
Rounding from real to integer is deterministic - the effect of rounding is defined in Ada 95 where a real number is exactly between two integers (see Incompatibility 16).  As the Ada 83 behaviour is implementation-defined, any program which required a deterministic result would have programmed around this problem.
Exact evaluation of static expressions in non-static contexts - in Ada 95, the evaluation is required to be exact.  In Ada 83, it is only required to be exact for universal expressions in a static context (see Incompatibility 22).  As most implementations use exact evaluation for all static expressions, most programs will be unaffected.
Numeric_Error renames Constraint_Error - in Ada 95, the declaration for Numeric_Error has been changed to a renaming of Constraint_Error (see Incompatibility 29).  If programs conform to recommended practice (handling Constraint_Error and Numeric_Error together) no effect will be noticed.
Evaluation order of defaulted generic actual parameters - the order of evaluation of defaulted generic actuals is arbitrary in Ada 95.  In Ada 83, defaulted generic actuals are evaluated after all supplied actuals (see Incompatibility 32).  It is unlikely that any normal program depends on the Ada 83 behaviour.
Upward Consistent Incompatibilities
If a legal Ada 83 compilation unit contains occurrences of any of the upward consistent incompatibilities, it will probably be an illegal Ada 95 compilation and hence the incompatibility would be detected at compile-time.  However, where an incompatibility is marked with a diamond (), most occurrences of the incompatibility will be detected at compile time, but it is theoretically possible (but very unlikely) that some occurrences will not be detected at compile time, but cause different behaviour at run time.
Upward Consistent Incompatibilities are classified as follows:
	Likely, Less likely and Unlikely Incompatibilities - for which it is likely, less likely or unlikely that an existing reasonably-sized Ada 83 program (50-100,000 lines) will exhibit occurrences of the incompatibility.  A much larger program is more likely to exhibit a less likely incompatibility.
	Implementation-Dependent Incompatibilities - where the definition of a new facility in Ada 95 may clash with an implementation-defined use in Ada 83, in which case occurrences of the incompatibility will probably cause a legal Ada 83 program to be an illegal Ada 95 program.
	Implementation-Choice Incompatibilities - where an Ada 83 facility is no longer defined in Ada 95, but where an implementation is free to continue supporting it.  Otherwise, use of the facility will cause a legal Ada 83 program to be an illegal Ada 95 program.
Likely Incompatibilities
Unconstrained generic actuals - in Ada 95, special syntax must be used in a generic formal parameter to allow unconstrained actuals (see Incompatibility 31).
Less likely Incompatibilities
New Keywords - in Ada 95, six new reserved keywords have been defined (see Incompatibility 2).
	Subtype Character has 256 positions - in Ada 95, subtype Character has 256 positions.  In Ada 83, it has 128 positions (see Incompatibility 5).
Definition of forcing occurrences tightened - in Ada 95, range constraints on a type after its declaration are treated as forcing occurrences.  In Ada 83 they are not (see Incompatibility 7).
Character literals always visible - in Ada 95, character literals are visible everywhere.  In Ada 83 they follow the usual rules of visibility (see Incompatibility 25).
Library package bodies illegal if not required - in Ada 95, it is illegal to provide a body for a library package that does not require one (see Incompatibility 28).  In Ada 83, it is allowed.
Numeric_Error renames Constraint_Error - in Ada 95, the declaration for Numeric_Error has been changed to a renaming of Constraint_Error (see Incompatibility 29).
Unlikely Incompatibilities
Bad pragmas illegal - in Ada 95, a pragma with an error in its arguments makes the compilation illegal.  In Ada 83, the pragma is ignored (see Incompatibility 1).
S'Base not defined for composite subtypes - in Ada 95, S'Base is not defined for a composite subtype S (see Incompatibility 3).
Wide_Character shares all character literals - as a result of adding subtypes Wide_Character and Wide_String to package Standard, Ada 83 character literals are always overloaded (see Incompatibility 6) and Ada 83 string literals are always overloaded (see Incompatibility 13).
Static matching of subtypes - in Ada 95, matching of subtypes is performed statically instead of at run-time (as in Ada 83) in array conversions (see Incompatibility 18) and generic instantiations (see Incompatibility 33).
Preference for universal numeric operators - in Ada 95, the overload resolution rules have been changed to simplify them and remove occurrences of the Beaujolais effect (see Incompatibility 19).
Explicit constraints illegal in uninitialised allocators designating access types - in Ada 95 it is illegal for an explicit constraint to be supplied for an uninitialised allocator if the subtype designates an access type.  In Ada 83, the constraint is ignored (see Incompatibility 20).
Exceptions in static expressions cause illegalities - in Ada 95 it is illegal for an exception to be raised whilst evaluating a static expression (at compile-time).  In Ada 83 the exception renders the expression non-static (see Incompatibility 21).
Functions returning local variables containing tasks - in Ada 95 it is illegal or Program_Error is raised if a function with a result type with a task subcomponent returns a local variable.  In Ada 83, it is erroneous to return a variable containing a local task (see Incompatibility 23).
Illegal to use value of deferred constant - in Ada 95 it is illegal to use the value of a deferred constant before it is set.  In Ada 83 it is erroneous (see Incompatibility 24).
Assume worst when checking generic bodies - Ada 83 generic contract-model violations have been overcome in Ada 95 by assuming the worst case in a generic body (see Incompatibility 30).
Definition of forcing occurrences tightened - in Ada 95, occurrences in pragmas are treated as forcing occurrences.  In Ada 83 they are not treated as forcing occurrences (see Incompatibility 34).
Illegal to change representation of types containing tasks - in Ada 95, it is illegal to give a representation item for a derived type containing a task (see Incompatibility 35).
	New identifiers added to package System - new identifiers in package System may introduce illegalities into a unit having a use clause for package System (see Incompatibility 36).
	New identifiers added to package Text_Io - new identifiers in package Text_Io may introduce illegalities into a unit having a use clause for package Text_Io (see Incompatibilities 37and 39).
	New identifiers added to package Standard - new identifiers in package Standard may clash with existing use-visible identifiers (see Incompatibility 40).
Unlikely Implementation-Dependent Incompatibilities
	New pragmas defined - the names of new pragmas may clash with implementation-defined pragmas (see Incompatibility 41).
	New attributes defined - the names of new attributes may clash with implementation-defined attributes (see Incompatibility 42).
New library units defined - the names of new (language-defined) library units may clash with user-defined or implementation-defined library units (see Incompatibility 43).
Unlikely Implementation-Choice Incompatibilities
Real attributes replaced - the Ada 83 attributes for a real subtype S (such as S'Mantissa) have been replaced by a different set in Ada 95 (see Incompatibility 10).
Certain pragmas removed - some pragmas have been removed from the language and pragma Priority has been moved to the Real-Time Annex (see Incompatibility 27).
Error Inconsistencies
Error inconsistencies only affect Ada 83 programs in which a predefined exception is implicitly raised.
Exceeding 'First or 'Last of an unconstrained floating point subtype - in Ada 95, the attributes S'First and S'Last of a floating point subtype S declared without a range constraint are treated as minimum bounds and may be exceeded without causing Constraint_Error (see Incompatibility 8).
Dependent compatibility checks performed on object declaration - In Ada 95, dependent compatibility checks are performed on object declaration.  Under certain circumstances in Ada 83, they are performed on subtype declaration (see Incompatibility 14).
Lower bound of concatenation changed for constrained array types - in Ada 95, the lower bound of the result of concatenation for a constrained array type is defined to be 'First of the index subtype.  In Ada 83, the lower bound of the result is 'First of the left hand operand, risking Constraint_Error (see Incompatibility 15).
Implicit array subtype conversion - Ada 95 allows sliding in more situations than did Ada 83, so Constraint_Error might not be raised as in Ada 83 (see Incompatibility 17).
Raising Time_Error deferred - in Ada 95, raising Time_Error can be deferred until Split or Year is called, or might not be raised at all.  In Ada 83, it is raised on "+" or "-" (see Incompatibility 26).
Input format for real numbers relaxed - in Ada 95, the format for real numbers accepted by Float_Io and Fixed_Io is relaxed, so that Data_Error will not be raised in some cases where it was raised in Ada 83 (see Incompatibility 38).
Detection and Avoidance of Incompatibilities
All upward consistent incompatibilities can be detected in Ada 83 source.  Detecting the error inconsistencies is not worthwhile.  The remaining inconsistencies are detectable with the exception of:
Real types may have less precision (see Incompatibilities 9 and 11), but the impact is small. 
Rounding from Real to Integer is deterministic (see Incompatibility 16), but the possibility of an occurrence of the incompatibility can be warned.  Avoidance is possible, but difficult.
Evaluation order of defaulted generic actual parameters (see Incompatibility 32), but the possibility of an occurrence of the incompatibility can be warned.  Avoidance is possible.
All the detectable incompatibilities identified can be avoided in Ada 83 source with the exception of:
Subtype Character has 256 positions (see Incompatibility 5), where avoidance is possible but can lead to obscure code. 
Real attributes replaced (see Incompatibility 10), which can only be avoided if the values yielded can be determined by alternative means.
Unconstrained Generic Actuals (see Incompatibility 31), which cannot be avoided at all.
Append_Mode added to File_Mode enumeration (see Incompatibility 37), where avoidance is possible can lead to obscure code.
Detailed Descriptions
This section lists all the known upward incompatibilities between Ada 83 and Ada 95.  They are sorted by section number in the Ada 83 Language Reference Manual [1].
The information for each incompatibility is grouped under the following headings:
(a)	Status
Whether the change is upward consistent, upward consistent if no exceptions or upward inconsistent and the likelihood of the incompatibility occurring in a normal Ada 83 program of a reasonable size and whether its occurrence can be avoided.
(b)	References
The LRM reference is to the Ada 83 Language Reference Manual [1].  The RM95 reference is to the Ada 95 Reference Manual [2].
(c)	Language Change
A description of the language change.
(d)	Rationale
The reason for the change.
(e)	Incompatibility
An explanation and example of any incompatibilities introduced.
(f)	Avoidance in Ada 83
A description of how to avoid the incompatibility in Ada 83 programs, or where this is not possible, how an Ada 83 program can be converted into a legal Ada 95 program with identical behaviour.  The objective is to achieve a single program with the same behaviour in Ada 83 and Ada 95.
(g)	Detection and Automated Correction
Whether the incompatibility is detectable and whether it is amenable to automatic conversion.  In this respect, a tool would not be expected to analyse a complete program and make "global" decisions, for example to change the name of a library package called Protected to one called Protected_1 (having checked that such a change would be upward compatible) and make all the consequential changes to units withing package Protected.  Note that any automated correction assumes that the program is a legal Ada 83 program.
(h)	Notes
Any relevant comments.
1.	Bad Pragmas Illegal
(a)	Status  	Upward Consistent - unlikely in normal programs - probably unintentional
(b)	References  LRM-2.8(9);  RM95-2.8
(c)	Language Change
In Ada 95, errors in a recognised pragma cause the pragma to be illegal.
(d)	Rationale
It is considered more useful to reject a bad pragma than to ignore it.
(e)	Incompatibility
In Ada 83, a recognised pragma with an error is ignored, for example:
type Bits is array (0 .. 31) of Boolean;
pragma Pack (Bit);						-- typo!
(f)	Avoidance in Ada 83
Remove or correct the pragma - it is extremely unlikely that the Ada 83 behaviour was intentional.
(g)	Detection and Automated Correction
Detection of a bad pragma is problematic, given that it can be implementation-defined.
(h)	Notes
Another change to pragmas in Ada 95 is that an implementation:
(1)	must have a mode in which unrecognised pragmas are warned,
(2)	may have a mode in which unrecognised pragmas are ignored,
(3)	may have a mode in which unrecognised pragmas are treated as errors
2.	New Reserved Words
(a)	Status  	Upward Consistent - less likely in normal programs - avoidable 
(b)	References  LRM-2.9(2);  RM95-2.9(2)
(c)	Language Change
Six new reserved words are introduced in Ada 95: abstract, aliased, protected, requeue, tagged and until.
(d)	Rationale
Alternatives which avoid new reserved words are the use of unreserved keywords (a new concept) and the use of combinations of existing keywords.  Neither of these options is preferable to the one-off inconvenience of the proposed incompatibility.
(e)	Incompatibility
Any Ada 83 program that uses any of these words as identifiers is an illegal Ada 95 program.  For example:
Protected : Boolean := False;
type Abstract is private;
procedure Requeue (The_Activity : Activity; On_Queue : Queue);
(f)	Avoidance in Ada 83
Avoid use of these six words as identifiers.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible but problematic.
3.	S'Base not defined for Composite Subtypes
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-3.3.3(9);
(c)	Language Change
In Ada 95, the attribute S'Base is not available for composite subtypes.
(d)	Rationale
In Ada 95, S'Base can be used as a subtype mark for elementary subtypes, particularly useful for numeric formal types in generics.  However, problems would arise if S'Base were allowed as a subtype mark for composite subtypes, for example:  
type T (X : Integer) is record
	A : String(1..X);
	B : String(1..X);
	C : Float;
end record;

type S is new T (80);
A legitimate representation for S would omit all overhead associated with the discriminant X, squeezing out any internal dope for A and B, since their upper bound is known at compile-time.  However, S'Base would then be unrepresentable.
In Ada 83, S'Base can only be used as a prefix to another attribute - as there are no interesting attributes for a composite subtype (as opposed to a composite object), there are unlikely to be Ada 83 programs that use S'Base for a composite subtype.
The only attributes available for a composite subtype are 'Constrained, 'First, 'Last, 'Length, 'Range and 'Size.  A generic is unlikely to contain occurrences of S'Base'Constrained, since the value is of no practical use.  Similarly, the value of S'Base'Size is of no practical use.  The other attributes are only applicable to constrained array subtypes.  S'Base for an array subtype is always unconstrained!
(e)	Incompatibility
An Ada 83 program that uses S'Base for a composite subtype S is an illegal Ada 95 program.  
(f)	Avoidance in Ada 83
Avoid using S'Base for a composite subtype S.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, if replacing S'Base by S is acceptable for composite subtypes.
4.	Derived Type Inherits all Operations of Parent
(a)	Status  	Upward Inconsistent but detectable - unlikely in normal programs - avoidable
(b)	References  LRM-3.4(11);  RM95-3.4(17)
(c)	Language Change
In Ada 95, a primitive operation of a type becomes derivable at the end of the declaration, unlike in Ada 83 where it only becomes derivable after the end of the visible part in which the type is declared.
(d)	Rationale
This change is introduced to ensure that a derived tagged type inherits all the operations of its parent type.  The change also applies to untagged types for consistency and simplification (it will probably fix more bugs than it creates), especially as in Ada 95 (unlike Ada 83) it is legal to derive from a type which is derived in the same visible part, so what is a rare situation in Ada 83 could be more common in Ada 95.
(e)	Incompatibility
(1)  An inconsistency can arise if a predefined operator is over-ridden, for example:
package P is
	type Some_Array is array (1 .. 10) of Integer;
	function "&" (Left, Right : Some_Array) return Some_Array;
	type New_Array is new Some_Array;
end P;
In Ada 83, the "&" operator for subtype New_Array is the default "&" for array subtypes and not that declared for subtype Some_Array.  In Ada 95, New_Array will inherit the "&" from its parent, Some_Array.
(2)  an (upward consistent) incompatibility can arise if a primitive operation is defined before the type is derived from, for example:
package P is
	type T is (A, B, C, D);
	function F (X : T := A) return Integer;
	type New_T is new T;
end P;
A call to P.F (B) or to P.F will resolve to the F for type T in Ada 83 (because New_T does not inherit F), but will be ambiguous in Ada 95 (because both T and New_T have a function F).
 (f)	Avoidance in Ada 83
If the (obscure) Ada 83 behaviour is intentional, the second type should be derived from the first type before the former's primitive operations, thus:
package P is
	type Some_Array is array (1 .. 10) of Integer;
	type New_Array is new Some_Array;
	function "&" (Left, Right : Some_Array) return Some_Array;
end P;
An alternative avoidance is to derive the two types from a common third subtype.
(g)	Detection and Automated Correction
Detection that an inconsistency may arise is possible.  Manual correction is necessary to determine whether the Ada 83 semantics need to be preserved.
5.	Subtype Character has 256 Positions
(a)	Status  	Upward Inconsistent but detectable - unlikely in normal programs - avoidable
		Upward Consistent - less likely in normal programs - difficult to avoid
(b)	References  LRM-3.5.2(1);  RM95-3.5.2(2)
(c)	Language Change
In Ada 95, subtype Character has 256 positions; in Ada 83, it has 128 positions.  Note that this change is also allowed in Ada 83 implementations.
(d)	Rationale
Although suitable for English-speaking nations, a character type based on ASCII is inappropriate for most of Europe.  ISO has defined a number of 256 character standards (such as Latin_1, Latin_2 etc).  This change accommodates non-English speaking nations.
(e)	Incompatibility
An Ada 83 program could be an illegal Ada 95 program if it has a case statement or an array indexed by Character, but it could be a legal Ada 95 program with different semantics if it relies on the position number or value of Character'Last.  For example:
type Char_Kind is (Numeric, Alphabetic, Other);
Kind_Array : array (Character) of Char_Kind :=			-- (1)
				('0' .. '9' => Numeric,
				 'A' .. 'Z' | 'a' .. 'z' => Alphabetic;
				 others => Other);
case Char is							-- (2)
	when Character'Val(0) .. Character'Val(63) => ...
	when Character'Val(64) .. Character'Val(127) => ...
end case;
I : Integer := Character'Pos(Character'Last);			-- (3)
Declaration (1) is legal in Ada 95 but probably requires changing. 
Statement (2) is illegal in Ada 95
Statement (3) will yield a different value in Ada 95.
(f)	Avoidance in Ada 83
As it is likely that allowing for 256 characters is outside the scope of the original requirement, avoidance is not the issue - a review of the requirements is necessary.
The upward inconsistency can be avoided by not depending on the position or value of Character'Last.  Avoiding the other incompatibilities is possible by treating the extra characters identically, via an others-choice.  However, this can lead to obscure code.
An alternative strategy that avoids all the above problems, but risks raising Constraint_Error on character input, is to replace all occurrences of Character by a subtype of Character covering the first 128 values.  However, in most cases such a change is unnecessary and leads to less clear code.
(g)	Detection and Automated Correction
Detection of the upward consistent incompatibilities is straightforward; detection that an inconsistency may arise is possible.  Manual correction is necessary to determine whether the required semantics of the program are those defined by Ada 95. No correction is necessary if the alternative strategy is adopted.
6.	All Character Literals Overloaded
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-3.5.2(1);  RM95-3.5.2(3)
(c)	Language Change
The Ada 95 version of package Standard defines a new character subtype, Wide_Character (see Incompatibility 5), whose literals include all of those defined for subtype Character.
(d)	Rationale
Subtype Wide_Character is required for non-alphabetic languages, such as Japanese Kanji and Chinese, where 256 literals are insufficient.
(e)	Incompatibility
All existing Ada 83 character literals are overloaded in Ada 95, so any expression or subtype conversion using only character literals is ambiguous and hence illegal, for example:
... Character ('x') ...
if 'x' = 'y' then ...
for I in 'A' .. 'Z' loop
(f)	Avoidance in Ada 83
One of the operands needs to be subtype qualified (or equivalent), for example:
... Character' ('x') ...
if Character'('x') = 'y' then ...
for I in Character range 'A' .. 'Z' loop
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by assuming the required subtype is Standard.Character as above.
(h)	Notes
WG9, the ISO Working Group with responsibility for maintaining the Ada standard, has decreed that this change can be introduced into Ada 83 compilers.
7.	Range Constraint on Scalar Type is Forcing Occurrence
(a)	Status  	Upward Consistent - less likely in normal programs - avoidable
(b)	References  LRM-3.5(5);  RM95-13.14(11)
(c)	Language Change
A range constraint on a scalar type (after its declaration) is a forcing occurrence (freezing point in Ada 95), so for example subsequent representation clauses for the type are illegal.
(d)	Rationale
For discrete subtypes, no evaluation of an expression of a type should be required prior to knowing the representation of the type.  If knowledge is required, then this occurrence becomes a freezing point for the type in Ada 95.
For real subtypes, AI-174 recognised that the values of 'First and 'Last depend on the representation chosen for the type.  The default representation can be overridden by a representation clause.  However, when an intervening subtype or derived type declaration occurs, the chosen representation is not available, so 'First and 'Last are not yet known.
(e)	Incompatibility
The following example is illegal in Ada 95, because the subtype declaration will be a forcing occurrence of Volt.
type Volt is delta 0.2 range 0.0 .. 255.0;
subtype Mini_Volt is Volt range 0.0 .. 0.9;
for Volt'Small use 0.125;
(f)	Avoidance in Ada 83
Ensure that a representation clause immediately follows the declaration to which it refers, a good practice for readability.
type Volt is delta 0.2 range 0.0 .. 255.0;
for Volt'Small use 0.125;
subtype Mini_Volt is Volt range 0.0 .. 0.9;
(g)	Detection and Automated Correction
Detection of the illegal representation (ie attribute definition) clause is straightforward.  Automated correction is possible, by moving the offending representation clause to precede the forcing occurrence.
8.	Exceeding 'First or 'Last of an Unconstrained Floating Point Subtype
(a)	Status  	Upward Consistent if no Constraint_Error - less likely in normal programs but 
			unlikely to require attention - avoidable
(b)	References  LRM-3.5.7(3);  RM95-3.5.7(11)
(c)	Language Change
In Ada 95, the attributes S'First and S'Last for an unconstrained floating point subtype S, specify the minimum range supported by the implementation for run-time operations on values of the corresponding numeric subtype.  This means that the range check (which could cause Constraint_Error to be raised) on assignment and parameter passing is optional provided the correct value is returned
An unconstrained floating point subtype arises from a floating point type declaration in which no range constraint is given.
(d)	Rationale
In Ada 83, an implementation is required to perform the range check on assignment and parameter passing but is allowed to avoid the range check during evaluation of an expression if the correct result is derived.  This could occur where an intermediate value is held in extended precision (which often gives extended range as well).  Without the rule, extended precision would have to be lost just to enable a (probably) useless range check to be made.  Note that if 'Machine_Overflows is False no overflow checks are made anyway!
In Ada 95 this permission is extended to variables.  Constraint_Error would only then be raised if such a variable were assigned to a location where a smaller range can be held.
(e)	Incompatibility
In the following examples, Constraint_Error is not guaranteed to occur:
type F is digits 6;
Data : Float := Float'Last + Float'Last;
Data2 : F := F'Last + F'Last;
 (f)	Avoidance in Ada 83
Do not rely on Constraint_Error being raised where expected.
(g)	Detection and Automated Correction
Detection of the incompatibility is not possible.  However, it is unlikely that any change is necessary, as few applications are likely to rely on a range check for bounds in excess of 1038.
Note that if identical Ada 83 behaviour is required, a possible (avoidance and) correction strategy is to introduce a constrained subtype, for example by replacing:
type F is digits 6;
with:
type F_Base is digits 6;
subtype F is F_Base range F'First .. F'Last;

9.	Floating Point Types may have Less Precision
(a)	Status  	Upward Inconsistent, Implementation-Dependent - unlikely - avoidable
(b)	References  LRM-3.5.7(4);  RM95-3.5.7(4) and G.2.1
(c)	Language Change
In the Ada 95 core language, the required accuracy of floating point arithmetic is undefined.  However in the Numerics Annex in strict mode,  the model numbers are defined in terms of the hardware radix.  In Ada 83, the safe and model numbers are defined in terms of a hypothesised binary radix.
(d)	Rationale
For the majority of applications, the accuracy model for floating point arithmetic is an irrelevance - the accuracy provide by the hardware will suffice.  For more demanding applications, the accuracy model is defined in the Numerics Annex.  It is likely that implementations will obey the rules in the annex anyway.  
The rules in the annex differ from those in Ada 83 to allow for implementations where the hardware radix is non-binary (such as some IBM hardware where the radix is 16) - the Ada 83 rules sometimes result in a double precision representation being required even though a single precision representation would satisfy the requirement for model numbers.  The change is seen as a desirable correction of an anomaly.
(e)	Incompatibility
In Ada 95, a floating point type could be represented in single precision when it was represented in double precision in Ada 83.  For example:
type T is digits Float'Digits range -Float'Last .. Float'Last;
With hexadecimal hardware, double precision is used in Ada 83, because Float has model and safe numbers with 21 binary digits in their mantissas, as is required to model the hypothesised hexadecimal hardware using a binary radix; thus Float'Last, which is not a model number, is slightly outside the range of safe numbers of the single precision type, making that type ineligible for selection as the representation of T even though it provides adequate precision.  In Ada 95, Float'Last (the same value as in Ada 83) is a model number and is in the safe range of Float on the hypothesised hardware, making Float eligible for the representation of T.
(f)	Avoidance in Ada 83
The lack of an accuracy model in the core language is unavoidable.  Even in implementations conforming to the Numerics Annex, for programs intended for hardware with a non-binary radix, do not use Float'Last (or numbers very close to it) in the range specification.
(g)	Detection and Automated Correction
In the general case, detection the inconsistency is not possible.  No correction is possible (or likely to be necessary).
For the special case described above, detection of is straightforward. Manual correction is necessary as it is likely that no correction is required - ie the inconsistency is generally benevolent.
(h)	Notes
In strict mode, implementations claiming conformance to the Numerics Annex have to provide guaranteed accuracy requirements.  In relaxed mode, no such guarantee need be made, just as for the core language.  Implementations must support both modes, though they need not differ.
10.	Real Attributes Removed
(a)	Status  	Upward Consistent, Implementation-Choice - unlikely in normal programs - 
			unavoidable if information required
(b)	References  LRM-3.5.8(4-13) and 3.5.10(4-12)
(c)	Language Change
In Ada 95, the attributes S'Mantissa, S'Emax, S'Large, S'Small, S'Epsilon, S'Safe_Emax, S'Safe_Large and S'Safe_Small for a floating point subtype S have been removed from the language.  A replacement set of attributes is defined in Ada 95 (in A.5.3).  
Similarly, the attributes S'Mantissa, S'Large, S'Safe_Large and S'Safe_Small for a fixed point subtype S have been removed from the language.  No replacements are defined in Ada 95.  
Implementations are encouraged to retain the Ada 83 attributes (as implementation-defined attributes) with the same values as in Ada 83.
(d)	Rationale
Given that the floating and fixed point models have been simplified, the original attributes have been replaced by alternatives with a better defined and more relevant meaning.  Given that some of the attributes differ slightly in Ada 95, it is preferable avoid an upward inconsistency by removing the attributes rather than redefining their values.
(e)	Incompatibility
An Ada 83 program containing these attribute names is an illegal Ada 95 program, unless an implementation chooses to retain them for upward compatibility.  Unless all do, portability will be compromised so avoidance/correction is advisable.
(f)	Avoidance in Ada 83
None, although it is unlikely that any existing programs (outside the Validation Suite) will use these attributes.  Also, if all implementations support these attributes, no change is required anyway!
(g)	Detection and Automated Correction
Detection of the use of these attributes is straightforward.  Manual correction is necessary.
11.	Fixed Point Types may have Less Precision
(a)	Status  	Upward Inconsistent, Implementation-Dependent - unlikely - avoidable
(b)	References  LRM-3.5.9(2);  RM95-3.5.9(8) and G.2.3
(c)	Language Change
In Ada 95, the values of a fixed point type are just the integer multiples of its small, and in the core language, the required accuracy of fixed point arithmetic is undefined.  The accuracy requirement for strict mode of execution is defined in the Numerics Annex.
In Ada 83, the model numbers are the integer multiples of its small, but extra values are allowed in between the model numbers.
(d)	Rationale
The change was made to simplify the fixed point number model.
(e)	Incompatibility
In Ada 83, an implementation was free to use extra bits available in the representation of a fixed point type for extra accuracy.  In Ada 95, this freedom is removed.  Alternatively, an Ada 95 implementation could make the same decision in Ada 95 as in Ada 83 by choosing a smaller default small than was allowed in Ada 83.  However, this would also result in an upward inconsistency (see Incompatibility 12).
(f)	Avoidance in Ada 83
If more accuracy than that provided in Ada 95 is required, then either provide an explicit value for Small, or choose a smaller value for Delta.
(g)	Detection and Automated Correction
Detection of the inconsistency is problematic - it depends on individual compiler behaviour and whether strict mode has been selected.  No correction is possible (or likely to be necessary).
(h)	Notes
In strict mode, implementations claiming conformance to the Numerics Annex have to provide guaranteed accuracy requirements.  In relaxed mode, no such guarantee need be made, just as for the core language.  Implementations must support both modes, though they need not differ.
12.	Default Small for a Fixed Point Subtype
(a)	Status  	Upward Inconsistent, Implementation-Choice but detectable - unlikely in 
			normal programs as implementations unlikely to exploit - avoidable
(b)	References  LRM-3.5.9 (5);  RM95-3.5.9(8)
(c)	Language Change
In Ada 95, the default Small for a fixed point subtype S is always equal to S'Base'Small, which is not necessarily the largest power of two not exceeding S'Delta.
(d)	Rationale
In Ada 83, a fixed point subtype S can have a different Small than its base type.  This is to allow implementations to hold a fixed point number left justified.  Ada 83 implementations are allowed to make S'Base'Small a smaller power of two (but not many compilers ever exploited this permission).  In Ada 95, an implementation can still make S'Base'Small smaller than necessary but the default Small for a fixed point subtype S must be the same as S'Base'Small.
(e)	Incompatibility
In Ada 83, the default value for Small for a fixed point subtype S is required to be the largest power of two not exceeding S'Delta.  In Ada 95, it could be a smaller power of two.  Hence the value of S'Small could be different in Ada 95.
(f)	Avoidance in Ada 83
Either provide an explicit value for Small or avoid using S'Small for a fixed point type S.  Use S'Delta if appropriate.
(g)	Detection and Automated Correction
Detecting occurrences of S'Small for a fixed point subtype S without a representation clause for Small is straightforward.  Automated correction is possible, by providing an attribute definition clause for S'Small (with a value of the largest power of two not exceeding S'Delta).
13.	All String Literals Overloaded
(a)	Status  Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-3.6.3(1);  RM95-3.6.3(2)
(c)	Language Change
The Ada 95 version of package Standard defines a new string subtype (Wide_String), whose literals include all of those defined for subtype String.
(d)	Rationale
Subtype Wide_String is required to enable strings of Wide_Characters to be formed (see Incompatibility 6).
(e)	Incompatibility
All existing Ada 83 string literals are overloaded in Ada 95, so any expressions using only string literals are ambiguous and hence illegal.  This situation is unlikely for manually-produced programs, but possible for automatically-generated programs, for example:
if "abc" = "xyz" then ...
(f)	Avoidance in Ada 83
One of the operands needs to be subtype qualified, for example:
if String' ("abc") = "xyz" then ...
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by assuming the required subtype is Standard.String.
(h)	Notes
WG9, the ISO Working Group with responsibility for maintaining the Ada standard, has decreed that this change can be introduced into Ada 83 compilers.
Calls to Text_Io operations are not affected as support for I/O of Wide_Strings is achieved via a dedicated package (Wide_Text_Io) rather than extra declarations in package Text_Io.
14.	Dependent Compatibility Checks Performed on Object Declaration
(a)	Status  	Upward Consistent if no Constraint_Error - very unlikely in normal programs - 
			avoidable
(b)	References  LRM-3.7.2(5);  RM95-3.7.1(10)
(c)	Language Change
In Ada 95, Dependent Compatibility Checks are deferred until object creation (if any).
(d)	Rationale
Performing the check on the subtype declaration is against the spirit of Ada - objects might never be declared.
(e)	Incompatibility
In Ada 83, Dependent Compatibility Checks are performed on subtype declaration, for example (assuming that J > 5).
subtype Sub1 is Integer range J .. 20;
subtype Sub2 is Integer range 1 .. 20;
type T1 (D1 : Sub1) is .....
type T2 (D2 : Sub2) is record
	S : T1 (D2);
end record;
subtype S2 is T2 (5);			-- Constraint_Error raised in Ada 83
X : S2;					-- Constraint_Error raised in Ada 95
In Ada 83, Constraint_Error is raised during the elaboration of S2 because the value of discriminant S2.T1 is 5, which is outside the range of Sub1.  The declaration for X would never be elaborated.
In Ada 95, Constraint_Error is raised during the elaboration of X because the discriminant for X.S is 5, which is outside the range of Sub1.
(f)	Avoidance in Ada 83
Do not rely on Constraint_Error being raised when expected, although it is difficult to imagine that existing programs depend on this behaviour!
(g)	Detection and Automated Correction
It is not worth detecting that a Constraint_Error might have been possible in Ada 83.
15.	Lower Bound of Concatenation Changed for Constrained Array Types
(a)	Status  	Upward Consistent if no Constraint_Error - unlikely in normal programs and 
			unlikely to require attention - avoidable
(b)	References  LRM-4.5.3(4);  RM95-4.5.3(6)
(c)	Language Change
In Ada 95, the lower bound of the result of concatenation for a constrained array type is defined to be 'First of the index subtype (as opposed to 'First of the left hand operand).
(d)	Rationale
In Ada 83, concatenation between objects of a constrained array type usually causes Constraint_Error to be raised, for example:
	type A is array (1..10) of Integer;			-- A is constrained array type
	X : A;
	X := X (6..10) & X (1..5);				-- raises Constraint_Error in Ada 83
	X := X (6) & X (7..10) & X (1..5);			-- OK, but obscure
In the first statement, because the lower bound of the left hand operand is "6", the lower bound of result of concatenation will also be "6", causing Constraint_Error to be raised (because 6..15 is not within the range of the index subtype).  
In Ada 95, constrained array types are not defined in terms of the equivalent unconstrained array type, and so it is reasonable for the rules for concatenation to differ for constrained and unconstrained array types.
(e)	Incompatibility
In cases such as those above, Constraint_Error will not occur in Ada 95, where it did in Ada 83.
(f)	Avoidance in Ada 83
Do not rely on Constraint_Error being raised when expected.
(g)	Detection and Automated Correction
It is not worth detecting where Constraint_Error might be raised in Ada 83.
16.	Rounding from Real to Integer is Deterministic
(a)	Status  	Upward Inconsistent, Implementation-Dependent but warnable - unlikely in well-		written programs, where the precise effect matters - avoidable with difficulty
(b)	References  LRM-4.6(7);  RM95-4.6(33)
(c)	Language Change
When converting a value from a real to an integer type, a value exactly between two integers is rounded away from zero.
(d)	Rationale
This is one of the most unnecessary implementation dependencies in Ada 83.  It should not be implementation-dependent whether Integer (1.0/2.0) yields 0 or 1.  
All other ISO languages that define a standard Round function define it to round away from zero.  Round-to-nearest-even (a possible alternative) is appropriate when converting from one floating point type to another, or when discarding the extra bits of precision developed during a floating point calculation, since (on a statistical basis) it produces more generally accurate results.  However, a conversion from floating point to integer is totally different.  
(e)	Incompatibility
In Ada 83, the choice of which integer to choose is left implementation-defined, for example:
Flt : Float := 0.5;
Int := Integer (Flt);
In Ada 83, the value of Int could be 0 or 1, depending on the implementation.  In Ada 95, the value of Int is always 1.
(f)	Avoidance in Ada 83
The implementation-dependence can be avoided by writing portable code, in which case no change of behaviour would occur in Ada 95.
(g)	Detection and Automated Correction
Detection of all conversions from real to integer is possible (but not very helpful).  Manual intervention would be necessary to achieve a specific rounding algorithm.
(h)	Notes
This language change is included in the list of upward incompatible changes, because some Ada 95 implementations will not be able to provide identical behaviour to their Ada 83 implementation.
In Ada 95, rounding-to-nearest-even can be obtained by use of the attribute S'Unbiased_Rounding.  Truncation can be obtained by use of the attribute S'Truncation.
17.	Implicit Array Subtype Conversion
(a)	Status  	Upward Consistent if no Constraint_Error - less likely in normal programs but 
		unlikely to require attention - avoidable
(b)	References  LRM-4.6(11);  RM95-4.6(37)
(c)	Language Change
In Ada 95, more implicit array subtype conversions (ie sliding) is allowed than in Ada 83, thus avoiding Constraint_Error.
(d)	Rationale
Such restrictions are unnecessary and counter-intuitive.
(e)	Incompatibility
In the following examples, Constraint_Error will not occur in Ada 95:
(1)	
type R1 is record
	S : String (1 .. 10);
end record;
type R2 is record
	S : String (11 .. 20);
end record;
V1 : R1;
V2 : R2 := R2' (S => V1.S);		-- Constraint_Error in Ada 83, slides in Ada 95
V3 : R2 := R2' (S => "abcde" & "efghi");	-- likewise, since 'First of concatenation is 1
(2)	parameter passing:
X : String (5 .. 7);
subtype String_1_3 is String (1 .. 3);
procedure P ( S : String_1_3);
P (X);				-- raises Constraint-Error in Ada 83, slides in Ada 95
(3)	an initialised allocator:
type A is access String (5 .. 7);
X : A := new String' ("abc");		-- raises Constraint-Error in Ada 83, slides in Ada 95
(f)	Avoidance in Ada 83
Do not rely on Constraint_Error being raised when expected.
(g)	Detection and Automated Correction
It is not worth detecting that a Constraint_Error might have been possible in Ada 83.
18.	Static Matching of Component Subtype in Array Conversions
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-4.6(11);  RM95-4.6(12)
(c)	Language Change
In Ada 95 the component subtypes in an array conversion must match statically - either the subtypes must both be static and matching, or they must be the same subtype.
(d)	Rationale
Reporting errors at compile-time (rather than run-time) is one of the objectives of the Ada 95 Revision - the incompatibilities are unlikely to occur.
(e)	Incompatibility
In Ada 83, the check is performed at run-time and Constraint_Error raised if the component subtypes did not match.
For example:
type A1 is array (1 .. 10) of Integer range I .. J;
type A2 is array (1 .. 10) of Integer range K .. L;
X1 : A1;
X2 : A2 := A2 (X1);
In Ada 83, Constraint_Error is raised unless I=K and J=L.In Ada 95, the conversion is illegal.
(f)	Avoidance in Ada 83
Ensure component subtypes match statically.  Code relying on the Ada 83 semantics is unnecessarily obscure.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Manual correction is necessary however.
(h)	Notes
There is a similar change for generic matching rules for formal array and access subtypes (see Incompatibility 33).
19.	Preference for Universal Numeric Operators
(a)	Status  	Upward Consistent - very unlikely in normal programs - avoidable.
(b)	References  LRM-4.6(15);  RM95-8.6(29)
(c)	Language Change
In Ada 95 preference is given to Universal Numeric operators during overload resolution, as opposed to preference against implicit conversion in Ada 83.
(d)	Rationale
The Ada 83 rule is difficult to understand; implementations have chosen different interpretations.  The official interpretation (by WG9's Ada Rapporteur Group) is so difficult to implement that the interpretation is not enforced.
The revised rules are simpler to understand, simpler to implement and eliminate Beaujolais effects, for example:
package P is
	procedure Q (B : Boolean);				-- Q1
end P;
with P; use P;
procedure Main is
	function "<" (X, Y : Integer) return Integer is ...
	procedure Q (I : Integer) is ...			-- Q2
begin
	Q (1 < 2);
end Main;
In Ada 83, the call to Q resolves unambiguously to a call on Q1 (because the call to Q2 involves implicitly converting the universal integer parameters to Integer), whereas if Q were not declared in package P, the call would resolve unambiguously to a call on Q2.  This is an example of the (pathological) Beaujolais effect - the adding or removing of a declaration has a semantic effect.
(e)	Incompatibility
In Ada 83, preference during overload resolution is against implicit conversion (see LRM-4.6(15)).  In Ada 95, this preference against has been removed.
If extra operators are provided for a numeric type then an Ada 83 program could be an illegal Ada 95 program, but only if it contained a Beaujolais effect.
In the above example, the call to Q is ambiguous in Ada 95.  If the declaration of P.Q or the use clause were removed, the call would resolve to a call on Q2 (as in Ada 83).
(f)	Avoidance in Ada 83
It is difficult to imagine that existing programs depend on this behaviour!
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by replacing the call to Q by a call to P.Q (in the example).
20.	Explicit Constraints Illegal in Uninitialised Allocators Designating Access Types
(a)	Status  	Upward Consistent - unlikely - avoidable
(b)	References  LRM-4.8(4);  RM95-4.8(4)
(c)	Language Change
In Ada 95 it is illegal to supply a explicit constraint (ie index or discriminant constraint) in an uninitialised allocator if the designated type is an access type.  In Ada 83, the constraint is ignored, ie it does not affect the subtype of the allocated object, according to AI-331.
(d)	Rationale
Making illegal a constraint that was ignored is a minor improvement to the language.
(e)	Incompatibility
The following example is illegal in Ada 95.  In Ada 83, the explicit constraint is ignored.
type String_Ptr is access String;
type String_Ptr_Ptr is access String_Ptr;
P: String_Ptr_Ptr := new String_Ptr (1..10);
 (f)	Avoidance in Ada 83
Do not supply explicit constraints for uninitialised allocators for a subtype that designates an access type.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible by removing the offending constraint.
21.	Exceptions in Static Expressions Cause Illegalities
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-4.9(2);  RM95-4.9(33)
(c)	Language Change
In Ada 95, if the evaluation of a static expression causes an exception (other than overflow), then the expression is illegal.  In Ada 83, to qualify as static, an expression must deliver a value (ie not cause an exception).  Note that in Ada 95, an implementation is required to evaluate such expressions exactly (see Incompatibility 22).
(d)	Rationale
The new rules make the evaluation of static expressions more portable - in Ada 83, it is implementation-dependent whether Constraint_Error is raised or the correct value is yielded.
Reporting errors at compile-time (rather than run-time) is one of the objectives of the Ada 95 Revision - only programs which contain logic to ensure that the expression is not evaluated at run-time will suffer.
(e)	Incompatibility
In the following examples, Constraint_Error will occur in Ada 83 and illegalities in Ada 95:
I := 999_999;			-- where I is an Integer and Integer'Last < 10**6
I := 1/0;				-- division by zero
Note that by encapsulating the offending expression into a suitable conditional statement, the exception can be avoided in Ada 83, but the evaluation will still be illegal in Ada 95, for example:
if Integer'Last >= 10**6 then
	declare
		I : Integer;
	begin
		I := 999_999;
	end;
end if;
Note also that in the following example, Constraint_Error may occur in Ada 83 but not in Ada 95:
type Short_Int is range -32768 .. 32767;
I : Short_Int := -32768;		-- run time evaluation in Ada 83 might cause Constraint_Error
(f)	Avoidance in Ada 83
Do not rely on Constraint_Error being raised during evaluation of static expressions.  Additionally, do not try to make code portable by techniques such as the conditional statement above.
(g)	Detection and Automated Correction
Detecting that an expression (raising an exception) would cause an illegality in Ada 95 is straightforward (it has to be detected by Ada 95 parsers).  Automated correction is possible, by replacing the static expression by a non-static expression, if this is the desired behaviour.
22.	Exact Evaluation of Static Expressions in non-Static Contexts
(a)	Status  	Upward Inconsistency, Implementation-Dependent and warnable - unlikely in 
			normal programs - avoidable
(b)	References  LRM-4.9(12) and 4.10(4);  RM95-4.9(25,28,29)
(c)	Language Change
In Ada 95, an implementation is required to evaluate static expressions in a non-static context exactly.  In Ada 83, an implementation can adopt a different strategy (for example, to yield the same result as execution on the target).
(d)	Rationale
Leaving the choice undefined is an unnecessary implementation dependency.
(e)	Incompatibility
In Ada 83, an implementation is allowed to evaluate static expressions in a non-static context exactly, but is free to evaluate them in a manner consistent with the target, for example:
F : constant Float := 1.0;
if F/3.0*3.0 = 1.0 then .....
In Ada 95, the value of the boolean expression is always true.  In Ada 83 it could be true or false.
(f)	Avoidance in Ada 83
The implementation-dependence can be avoided in Ada 83 by ensuring that such expressions only occur in a static context, for example:
F : constant Float := 1.0;
G : constant Float := F/3.0*3.0;
if G = 1.0 then .....
(g)	Detection and Automated Correction
Detection of all static expressions in non-static contexts is possible.  Manual intervention is necessary to determine whether a different value might be obtained in Ada 95.
(h)	Notes
This language change is included in the list of incompatible changes, because some Ada 95 implementations will not be able to provide identical behaviour to their Ada 83 implementation.
23.	Functions returning Local Variables containing Tasks
(a)	Status  Upward Consistent - very unlikely in normal programs - avoidable
(b)	References  LRM-5.8;  RM95-6.5(17)
(c)	Language Change
A function with a limited by-reference result type (ie a task, protected or Limited_Controlled type or a type having a subcomponent of one of these types) and which attempts to return a local variable or a formal parameter is illegal or raises Program_Error.  In general, the check can be performed at compile time but within a generic body this may not be possible.
(d)	Rationale
In Ada 83, the only types that become limited by-reference types in Ada 95 are task types and those with task subcomponents.  AI 867 defines that if a function returns a local task, further execution is erroneous.  It is highly desirable to remove this erroneousness in Ada 95 especially as more types are limited by-reference types in Ada 95.  Although this rule might trap a few Ada 83 programs which were not erroneous, alternative rules for removing the erroneousness are less acceptable.
(e)	Incompatibility
In Ada 83, according to AI 867, if a function returns a locally declared task, further execution is erroneous.  In Ada 95 the following cases are illegal, but were legal and not erroneous in Ada 83:
(1)	returning a formal parameter containing a task,
(2)	returning a local variant record whose current variant does not include a task but which has a task in some other variant, for example:
task type Tt is ....
type L (D : Boolean := False) is record
	case D is
		when False => I : Integer;
		when True => T : Tt;
	end case;
end record;
function F return L is
	Local_L : L ;
begin
	return Local_L;
end F;
(f)	Avoidance in Ada 83
Do not return local variables or formal parameters containing tasks from functions.
(g)	Detection and Automated Correction
Detection of functions returning local variables or formal parameters containing tasks is possible except in generic bodies where an instantiation would be illegal only because of the body - a violation of the contract model.  Manual correction is necessary anyway.
24.	Illegal to use Value of Deferred Constant
(a)	Status  	Upward Consistent - very unlikely in normal programs - avoidable
(b)	References  LRM-7.4.3(4);  RM95-7.4(9)
(c)	Language Change
In Ada 95, trying to use the value of a deferred constant (before it has been set) is illegal.
(d)	Rationale
In Ada 95, deferred constants are more important than in Ada 83: the value can be supplied via a Pragma Import (useful when interfacing to other languages); a child package may wish to export a constant of a private type exported by its parent; deferred constants can be of any type (not just private types) so will be more frequently used.  The change is a ramification of the revision to forcing occurrences (see Incompatibility 7).  Also, reporting errors at compile-time (rather than treating the program as erroneous) is one of the objectives of the Ada 95 Revision.
(e)	Incompatibility
In Ada 83, trying to use the value of a deferred constant (before it has been set) is erroneous.  An implementation is free to raise Program_Error if it can detect this situation.  
In the following example, the elaboration of Storage requires that Null_Data be already elaborated:
package Adt is
	type Data is private;
	Null_Data : constant Data;
private
	type Data is range 0 .. 255;
	type Hidden_Data (D : Data := Null_Data) is record .... end record;
	Storage : Hidden_Data;
	Null_Data : constant Data := 0;
end Adt;
(f)	Avoidance in Ada 83
Do not access the value of a deferred constant before its value has been set.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward (it has to be detected by Ada 95 parsers).  Automated correction is possible by moving the offending declaration to after the full constant declaration.
25.	Character Literals Always Visible
(a)	Status  Upward Consistent - less likely in existing programs - avoidable
(b)	References  LRM-8.7(1);  RM95-4.2(3)
(c)	Language Change
In Ada 95, character literals are resolved using context, rather than content.
(d)	Rationale
The new rule makes character literals consistent with numeric, null and string literals - they are usable everywhere; string literals are more usable.  In Ada 83, visibility of character literals cannot be used to help resolve the type of a string literal, but they nevertheless must be directly visible.  For example:
with Ebcdic;
package Ebcdic_Strings is
	subtype E_Character is Ebcdic.Character;
	type E_String is array (Positive range <>) of E_Character;
end Ebcdic_Strings;
----------------------------------------------
with Ebcdic_Strings; 
package Const is
	M : constant Ebcdic_Strings.E_String := "this is a string" & " and so is this";
end Const;
The above is illegal in Ada 83 and Ada 95.  In Ada 95, inserting "use type Ebcdic_Strings.E_String" makes "&" directly visible, but unless character literals were usable anywhere, a use-clause for package Ebcdic would be necessary.  This would then be the only case where the usability of an operation of one type (namely a string literal) depended on the direct visibility of some other type declaration.  
(e)	Incompatibility
In Ada 83, during overload resolution, only those character literals directly visible are considered.  In Ada 95, context must determine the character type, for example:
with Ebcdic;
procedure Test is
	procedure Put (C : Character) is ...;
	procedure Put (C : Ebcdic.Character) is ...
begin
	Put ('A');
end Test;
In Ada 83, 'A' resolves to Standard.'A', but in Ada 95 it is ambiguous because there are two candidate Put procedures: one which takes a parameter of subtype Character and another that takes a parameter of subtype Ebcdic.Character - the character 'A' could be either of these because Ebcdic.Character is visible.
(f)	Avoidance in Ada 83
The character literal needs to be subtype qualified, for example:
	Put (Character'('A'));
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by qualifying the literal with the appropriate subtype.
26.	Raising Time_Error Deferred
(a)	Status  	Upward Consistent if no Time_Error - unlikely in normal programs and 
			unlikely to require attention - avoidable
(b)	References  LRM-9.6(6);  RM95-9.6(26)
(c)	Language Change
In Ada 95, Time_Error need only be raised on "+" or "-" if the time is unrepresentable (ie overflow occurs).  Instead it is raised in procedures Year and Split if the year number is out of range (of subtype Year_Number).
(d)	Rationale
This change simplifies the implementation of "+" and "-" by removing the range check (only an overflow check need be made).  It also allows consideration of local time zone information to be delayed until Year or Split.
(e)	Incompatibility
In a program for which Time_Error was raised during a call to "+" or "-" in Ada 83, in Ada 95, raising the exception would be deferred until Year or Split were called or not raised at all.
	T : Time := Time_Of (2099, 12, 31); 	 --last day of calendar
	T := T + Day_Duration'Last;	--Time_Error in Ada 83 (2100, 1, 1)
	Y : Year_Number := Year (T);	--Time_Error in Ada 95 (Year > 2099)
In Ada 83, Time_Error is raised on the "+" because the year number is out of range.  In Ada 95, this check is deferred until the need to represent the year number, ie in a call to Year or Split.
(f)	Avoidance in Ada 83
Do not rely on Time_Error being raised.
(g)	Detection and Automated Correction
It is not worth detecting that a Time_Error might have been possible in Ada 83.
27.	Certain Pragmas Removed
(a)	Status  	Upward Consistent, Implementation-Choice - unlikely to cause incompatibility
(b)	References  LRM-9.8(1), 9.11(9), 13.7(4), 13.9(1)
(c)	Language Change
Pragmas Interface, Memory_Size, Shared, System_Name and Storage_Unit are undefined in the Ada 95 language; pragma Priority is undefined in the Ada 95 core language.
(d)	Rationale
Pragma Priority has been moved to the Real-Time Annex - forcing implementations to give an illusion of supporting priorities is counter-productive.
Pragma Shared has been replaced by pragma Atomic, which is defined in the Systems Programming Annex.
Pragmas System_Name, Storage_Unit and Memory_Size were introduced in Ada 83 as a mechanism for changing the values of the corresponding named numbers in package System.  Some Ada 83 implementations allow package System to be recompiled, so these pragmas are unnecessary.  Other implementations neither allowed package System to be recompiled nor supported the pragmas.  Only those implementations that did not allow package System to be recompiled but supported the pragmas will be affected.  These implementations are likely to continue to support the pragmas.  
Pragma Interface has been replaced by pragma Import, with additional functionality.
Leaving the pragmas in the language just to avoid a short-term transition problem would be short-sighted.  The language will be simpler without them, and implementations are allowed to continue to support them.
(e)	Incompatibility
These pragmas are defined in Ada 83.  If an Ada 95 implementation does not support them, occurrences of the pragma will be ignored, the program will still be legal (except for pragma Interface), but the intended effect will not happen.  However, implementations which support the required functionality are unlikely to remove support for the Ada 83 pragmas.
(f)	Avoidance in Ada 83
Use of pragma Priority in a program not requiring support for the Real-Time Annex is unlikely.
Implementations that fully supported pragma Shared are likely to continue to do so.
Use of the System-related pragmas is unlikely to occur in normal library units - they are usually restricted to being compiled in empty libraries.
Use of pragma Interface cannot be avoided (if the functionality is needed).
(g)	Detection and Automated Correction
Detection of the use of the pragmas in straightforward.  Automated correction is possible for pragmas Interface and Shared - by changing them to pragma Import or Atomic (as appropriate).  For the other pragmas, no change is necessary - any warnings generated by an Ada 95 compiler would need to be addressed.
 (h)	Notes
The named number declarations in package System with the same names as the System-related pragmas have not been removed.
28.	Library Package Bodies Illegal if not Required
(a)	Status  	Upward Consistent - less likely (but risky) in normal programs - avoidable
(b)	References  LRM-10.1;  RM95-7.2(4)
(c)	Language Change
In Ada 95, it is illegal to provide a body for a library package that does not require one.
(d)	Rationale
This avoids a nasty and not so rare error in Ada 83 - if a body is provided for a library package that does not need one, then if the package specification is subsequently changed, the body becomes obsolete but, as it is optional, subsequent builds incorporating the package will not incorporate the body, unless it is manually recompiled.
(e)	Incompatibility
In Ada 83, a body can be provided for a package that does not need one; in Ada 95, such a body would be illegal, for example:
package P is
	Global_Variable : Integer;
end P;
-----------------------------------------
package body P is
	function Func return Integer is ...;
begin
	Global_Variable :=Func;
end P;
A similar situation can arise where an optional package body declares tasks.
(f)	Avoidance in Ada 83
Avoid optional bodies by declaring a spurious procedure or incomplete type in the private part of the package, for example:
package P is
	Global_Variable : Integer;
private
	type Body_Required;
end P;
-----------------------------------------
package body P is
	type Body_Required is (Dummy);
	function Func return Integer is ...;
begin
	Global_Variable :=Func;
end P;
(g)	Detection and Automated Correction
Detection of a non-required library package body is straightforward.  Automated correction, by changing its specification may be unwise, due to the non-uniform nature of program (sub)libraries.
(h)	Notes
In Ada 95, the illegality can be avoided by adding a pragma Elaborate_Body to the package specification - this makes the body required and causes it to be elaborated immediately after the specification.
29.	Numeric_Error renames Constraint_Error
(a)	Status  	Upward Inconsistent but detectable - unlikely in well-written normal programs as 
			Ada 83 behaviour is not well defined - avoidable
		Upward Consistent - less likely in well-written normal programs as Ada 83 
			behaviour is not well defined - avoidable
(b)	References  LRM-11.1(6);  RM95-J.6(2)
(c)	Language Change
In Ada 95, the exception Numeric_Error is a renaming of Constraint_Error, instead of being a distinct exception.
(d)	Rationale
In Ada 83, it was not always clear whether Numeric_Error or Constraint_Error should be raised in certain circumstances, so the ARG issued a non-binding interpretation that allowed an implementation to raise Constraint_Error wherever Numeric_Error was specified.
It is less disruptive to existing Ada 83 programs to make Numeric_Error a renaming of Constraint_Error, rather than remove it from the language (as once proposed).
(e)	Incompatibility
If a frame has an explicit handler for Numeric_Error or Constraint_Error but not both, an occurrence of the other exception would be caught by that handler in Ada 95 but not in Ada 83, a change in behaviour, for example:
exception
	when Numeric_Error => Action_1;		-- also catches Constraint_Error in Ada 95
end;
exception
	when Constraint_Error => Action_2;		-- also catches Numeric_Error in Ada 95
	when others => Action_1;			-- catches Numeric_Error in Ada 83
end;
If a frame has different explicit handlers for Numeric_Error and Constraint_Error, the program will be illegal in Ada 95, for example:
exception
	when Numeric_Error => Action_1
	when Constraint_Error => Action_2;
end;
Note that a further inconsistency will occur in programs that explicitly raise Numeric_Error.
(f)	Avoidance in Ada 83
Always handle Numeric_Error and Constraint_Error in the same handler (sensible for portability), for example:
when Numeric_Error | Constraint_Error => Some_Action;
Note, this is legal in Ada 95, due to a language change that allows multiple views of an exception to be choices in the same handler.
(g)	Detection and Automated Correction
Detection is possible of situations in which Numeric_Error and Constraint_Error would be handled differently.  Manual correction is necessary to ensure the required behaviour .
30.	Assume Worst when Checking Generic Bodies
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-12.1;  RM95-12.3(11)
(c)	Language Change
In Ada 95, an array-type formal parameter of a generic formal subprogram is assumed to be unconstrained, precluding the use, in the generic body, of an others choice for an aggregate as an actual parameter in a call to the formal subprogram.
(d)	Rationale
In Ada 83, it is possible that the legality of a generic instantiation depends on the contents of the generic body, with the implication that changing the body could render existing (legal) instantiations illegal!  This change eliminates that possibility.  See also Incompatibility 31.
(e)	Incompatibility
An Ada 83 generic body in which an aggregate is supplied as an actual array-type parameter in a call to a generic formal subprogram, is illegal in Ada 95 if an others choice is supplied.  This is because the actual subprogram in the instantiation could have a corresponding unconstrained array subtype (the parameters are only required to have the same base type, not the same subtype!), for example:
subtype Small_String is String (1 .. 255);
generic
	with procedure Gp (S : Small_String);
package Gen_Pack is ....;
-----------------------------------------------------------
package body Gen_Pack is
	....
	Gp ((others => "*"));					-- illegal in Ada 95
	....
end Gen_Pack;
-----------------------------------------------------------
procedure P1 (S : Small_String);
procedure P2 (S : String);
package Pack_1 is new Gen_Pack (P1);				-- legal in Ada 83
package Pack_2 is new Gen_Pack (P2);				-- illegal in Ada 83
In Ada 83, the instantiation for Pack_2 is illegal because the body of Gen_Pack calls Gp (ie calls P2) with an aggregate with an others choice, which is illegal because P2 has a parameter of an unconstrained array subtype (viz String).  In Ada 95, the body of Gen_Pack is illegal.
(f)	Avoidance in Ada 83
Qualify the aggregate with the subtype name:
	Gp (Small_String'(others => "*"));
(g)	Detection and Automated Correction
Detection of the incompatibility in the body is straightforward.  Automated correction is possible by qualifying the aggregate with the subtype name.
31.	Unconstrained Generic Actual Subtypes
(a)	Status  	Upward Consistent - very likely in normal programs - unavoidable
(b)	References  LRM-12.1.2;  RM95-12.5.1(6)
(c)	Language Change
Ada 95 provides new syntax for a generic formal private type to indicate that the actual subtype is allowed to be indefinite (ie unconstrained without defaults).  The old syntax is retained, but the meaning is changed to require definite actuals (ie constrained or with defaults).
 (d)	Rationale
In Ada 83, no indication is given in a generic formal type declaration as to whether the actual needs to be constrained, for example because the body declares a variable of the subtype.  It is thus possible for a legal instantiation to become illegal if the body is changed.  
The alternative of introducing a pragma (such as Definite) was considered an inferior solution from a language design perspective.
(e)	Incompatibility
An Ada 83 program is an illegal Ada 95 program if an indefinite subtype is used as a generic actual parameter.  For example the following legal Ada 83 program is illegal in Ada 95 (because String is an indefinite subtype).
generic
	type Element_Type is private;
package Stack is ....
-----------------------------------------------------------
with Stack;
package String_Stack is new Stack (Element_Type => String);
 (f)	Avoidance in Ada 83
None.  It is recommended that formal private types be annotated with an appropriate comment, thus:
generic
	type Element_Type is private;  -- !! (<>) in Ada 95
package Stack is ... 
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward but does not address future instantiations.  
A possible strategy for the (automatic) conversion of an Ada 95 program is to change each occurrence of a generic formal private type to be indefinite, and then check the corresponding body is still legal, for example:
generic
	type Element_Type (<>) is private;
package Stack is .... 
 (h)	Notes
WG9 has recommended that Ada 83 compilers be allowed to accept the new syntax (with or without any semantic effect), in which case lack of use of the new syntax must have no semantic effect.
Some predefined library units in Ada 83 have been changed, such as Unchecked_Conversion, Unchecked_Deallocation and Sequential_Io (but not Direct_Io).
32.	Evaluation Order of Defaulted Generic Actual Parameters
(a)	Status  	Upward Inconsistent, Implementation-Choice but warnable - unlikely and 
			non-portable - avoidable
(b)	References  LRM-12.3(17);  RM95-12.3(20)
(c)	Language Change
In Ada 95, the supplied and defaulted actual parameters in a generic instantiation are evaluated in an arbitrary order consistent with any dependencies of defaults on other parameters.  In Ada 83, defaulted parameters are evaluated after all supplied parameters.
(d)	Rationale
There are no obvious benefits to the existing rule.  It is an unnecessary and arbitrary restriction, preventing a natural order of evaluation.
(e)	Incompatibility
In Ada 83, the order of evaluation of the supplied actual parameters in a generic instantiation is arbitrary, this must then be followed by the evaluation of the default values of any omitted generic associations (if any).  It is likely that not many users are aware of this behaviour let alone rely on it!
(f)	Avoidance in Ada 83
Do not rely on the order of evaluation of generic parameters.  Relying on a particular order is unnecessarily obscure.
(g)	Detection and Automated Correction
Detection that an inconsistency may arise is possible.  Manual correction is necessary to determine whether the required semantics of the program (ie the Ada 83 defined order) would be met by the Ada 95 semantics.  Note that relying on a particular order is a bounded error.
(h)	Notes
An implementation can avoid the incompatibility by retaining Ada 83 semantics, but any program that relied upon the Ada 83 semantics would be non-portable in Ada 95.
33.	Compatibility Checks at Compile-Time
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-12.3.2(3), 12.3.4(4), 12.3.5(1);  RM95-12.5.1(14), 12.5.3(6,7), 12.5.4(3)
(c)	Language Change
In Ada 95, the following checks for matching of actual and formal generic parameters are performed at compile-time:
(1)	for formal private types, that the discriminant subtypes match,
(2)	for formal array types, that the component subtypes match,
(3)	for formal array types, that the index subtypes or ranges match,
(4)	for formal access types, that the designated subtypes match, 
(d)	Rationale
Reporting errors at compile-time (rather than run-time) is one of the objectives of the Ada 95 Revision.  However, the incompatibilities are unlikely to occur in practice.
(e)	Incompatibility
In Ada 83, the checks are performed at run-time and Constraint_Error raised if the subtypes do not match.
For example:
subtype S1 is Integer range K .. L;
generic
	type F (D : S1) is private;
package P is ....
type S2 is Integer range I .. J;
type A (D : S2) is record ... end record;
package Pp is new P (A);
In Ada 83, Constraint_Error is raised unless I=K and J=L.In Ada 95, the instantiation is illegal.
(f)	Avoidance in Ada 83
Ensure subtypes match statically.  Code relying on the Ada 83 semantics is unnecessarily obscure.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Manual correction is necessary however.
(h)	Notes
See Incompatibility 18 for another example of static matching.
34.	Occurrences within Pragmas can be Forcing Occurrences
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-13.1;  RM95-13.14
(c)	Language Change
In Ada 95, occurrences within pragmas can be forcing occurrences.
(d)	Rationale
It is likely that the Ada 95 behaviour will fix more bugs than it will cause given the Ada 83 behaviour.
(e)	Incompatibility
In Ada 83, occurrences within pragmas that would otherwise be forcing occurrences cause the pragma to be ignored!  
In the example below, the occurrence of Short_Int'Size in a pragma should be a forcing occurrence but as it occurs inside a pragma, this is not allowed, so the pragmas is ignored.
package P is
	type Short_Int is range 1 .. 10;
	task T is
		pragma Priority (Short_Int'Size);
	end T;
private
	for Short_Int'Size use 8;
end P;
(f)	Avoidance in Ada 83
Ensure that a pragma does not precede any representation clauses for the type that would be frozen.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by moving such pragmas further down the declarative part, or by moving the representation clause to immediately follow the type definition.  Because of placement restrictions on some pragmas, the latter action may not be possible, for example:
package P is
	type Short_Int is range 1 .. 10;
	for Short_Int'Size use 8;
	task T is
		pragma Priority (Short_Int'Size);
	end T;
private
	...
end P;
35.	Illegal to Change Representation of Types containing Tasks
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-13.6;  RM95-13.1(10)
(c)	Language Change
In Ada 95, it is illegal to give a representation item for a derived type which is a limited by-reference type.
(d)	Rationale
As limited by-reference types are required to be passed by reference and returned by reference, it is impossible to change the representation as part of parameter passing.  And, since assignment is not defined, any kind of representation change is impossible
(e)	Incompatibility
In Ada 83, the only types that become limited by-reference types in Ada 95 are task types and those with task subcomponents, for example:
task type T;
type By_Ref is record
	T_Comp : T;
end record;
type New_By_Ref is new By_Ref;
pragma Pack (New_By_Ref);
In Ada 95, the pragma is illegal.
(f)	Avoidance in Ada 83
Avoidance changing the representation of types with subcomponents of a task type.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Manual correction is necessary as the only possible correction is to apply the representation item to the parent subtype.
36.	New Identifiers Added to Package System
(a)	Status  	Upward Inconsistent, Implementation-Dependent but detectable - unlikely in 
			normal programs - avoidable
		Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-13.7;  RM95-13.7(3-18)
(c)	Language Change
New identifiers have been added to package System.
(d)	Rationale
The alternative of adding the identifier to a child package of System has been used in many cases, but this is not considered appropriate in all cases.
(e)	Incompatibility
	If an Ada 83 program has a use-clause for package System and has use-visibility of any of these new identifiers (via another package), the program is an illegal Ada 95 program.  
	package Other_Package is
		Word_Size : constant Integer := 32;
	end Other_Package;
	use System; use Other_Package;
	... Word_Size ...		-- in Ada 83 refers to Other_Package.Word_Size
					-- in Ada 95 neither are visible
	If an implementation has used any of these identifiers in package System in a similar way, it is possible that the program is a legal Ada 95 program with different semantics!  
The following new identifiers are defined in package System:
(1)	the numbers: Max_Base_Digits (3.5.7), Max_Binary_Modulus (3.5.4), Max_Nonbinary_Modulus (3.5.4), Word_Size (13.7);
(2)	the constants: Null_Address (13.7), Default_Bit_Order (13.7), Default_Priority (H.1);
(3)	the subtypes: Any_Priority (H.1), Bit_Order (13.5.2), Interrupt_Priority (H.1).
(4)	the enumeration literals: High_Order_First (13.7), Low_Order_First (13.7).
(f)	Avoidance in Ada 83
Do not include use-clauses for package System.  This is good practice as implementations have always been free to add identifiers to package System.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  
If the collision occurs with an identifier (Word_Size) declared in another package (Other_Package), then automated correction is possible, by replacing Word_Size by Other_Package.Word_Size.
If the implementation used the identifier in its version of System, manual correction is necessary.
37.	Append_File Added to File_Mode Enumeration
(a)	Status  	Upward Inconsistent but detectable - unlikely in normal programs - avoidable
		Upward Consistent - unlikely in normal programs - unavoidable
(b)	References  LRM-14.1(8);  RM95-A.8.1(4) and A.10.1(4)
(c)	Language Change
In Ada 95, enumeration subtype File_Mode in packages Sequential_Io and Text_Io has an additional literal (Append_File).
(d)	Rationale
This is the most appropriate way to add append-to-file functionality.
(e)	Incompatibility
An Ada 83 program could be an illegal Ada 95 program if it had a case statement or an array indexed by File_Mode, but it could be a legal Ada 95 program but with different semantics if it relied on the position number of File_Mode'Last, for example:
Mode_Array : array (Text_Io.File_Mode) of Boolean := (In_File => True,		-- (1)
							         Out_File => False);
case Mode is									-- (2)
	when In_File => ...
	when Out_File => ...
end case;

if M = Out_File then
	-- some action for out files
else
	-- some action for in files							-- (3)
end if; 

for M in Text_Io.File_Mode loop
	Text_Io.Put ("*");								-- (4)
end loop;
Declaration (1) is illegal in Ada 95.
Statement (2) is illegal in Ada 95.
Statement (3) will be executed for append files in Ada 95
Statement (4) will have a different effect in Ada 95.
This change has an additional incompatibility for programs that have use clauses for package Text_Io.  For further details, see Incompatibility 36 for a similar problem in package System.
(f)	Avoidance in Ada 83
The upward inconsistency can be avoided by not depending on the value of File_Mode'Last.  The other incompatibilities can be avoided by using an others choice to specify the required behaviour for Append_File.  However, such a change leads to obscure code.
(g)	Detection and Automated Correction
Detection of the upward consistent incompatibilities is straightforward; detection that an inconsistency may arise is possible.  Manual correction is necessary to determine whether the required semantics of the program are those defined by Ada 95.
38.	Input Format for Real Numbers Relaxed
(a)	Status  	Upward Consistent if no Data_Error - unlikely in normal programs and 
			unlikely to require attention - avoidable
(b)	References  LRM-14.3.5(10);  RM95-A.10.6(6)
(c)	Language Change
In Ada 95, Float_IO and Fixed_IO will accept any numeric input that conforms to ISO 6093, which includes numbers with no leading digit, no trailing digit or no decimal point.
(d)	Rationale
The new rule makes it possible to input data files produced by programs written in other languages (such as Fortran)
(e)	Incompatibility
In Ada 83, the only acceptable input format for real numbers is to include: a leading digit, a decimal point and a trailing digit.  Thus, a program that inputs a number that is acceptable in the Ada 95 rules but not in the Ada 83 rules (such as .75), will cause the exception Data_Error to be raised as an Ada 83 program but will execute without such an exception as an Ada 95 program.
(f)	Avoidance in Ada 83
Do not rely on Data_Error being raised.
(g)	Detection and Automated Correction
Detection is not possible, but the need for a correction is extremely unlikely.
(h)	Notes
The Ada 95 attribute S'Value for a real subtype S will also accept real numbers conforming to the relaxed syntax.
39.	New Identifiers Added to Package Text_Io
(a)	Status  	Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-14.3.10;  RM95-A.10.1(13-21, 43-45)
(c)	Language Change
New identifiers have been added to package Text_Io.
(d)	Rationale
The alternative of adding the identifier to a child package of Text_Io has been used in many cases, but this is not considered appropriate in all cases.
(e)	Incompatibility
If an Ada 83 program has a use-clause for package Text_Io and has use-visibility of any of these new identifiers (via another package), the program is an illegal Ada 95 program.  
	package Other_Package is
		type File_Access is ...;
	end Other_Package;
	use Text_Io; use Other_Package;
	... File_Access ...		-- in Ada 83 refers to Other_Package.File_Access
					-- in Ada 95 neither are directly visible
The following new identifiers are defined in package Text_Io:
(1)	the subprograms: Set_Error, Standard_Error, Current_Error, Flush, Is_Open, Look_Ahead, Get_Immediate;
(2)	the generic packages: Modular_Io, Decimal_Io;
(3)	the subtype: File_Access;
 (f)	Avoidance in Ada 83
Do not apply a use-clause for package Text_Io.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward. Automated correction is possible, by replacing (in the example) File_Access by Other_Package.File_Access.  
An alternative strategy is to remove the use-clause for package Text_Io (and/or Other_Package), replacing it with a use-type-clause if necessary, and adjusting all references to identifiers declared directly within it.
 (h)	Notes
The identifier Append_File has also been added to package Text_Io (see Incompatibility 37)
40.	New Identifiers in Package Standard
(a)	Status  	Upward Inconsistent but detectable - unlikely in normal programs - avoidable
		Upward Consistent - unlikely in normal programs - avoidable
(b)	References  LRM-C;  RM95-A.1(36,41)
(c)	Language Change
New identifiers have been added to package Standard: 
Wide_Character and Wide_String (see Incompatibilities 6 and 13).
(d)	Rationale
Although such identifiers have been kept to a minimum, there are cases where identifiers need to be directly visible for a new feature to be readily usable.
(e)	Incompatibility
If an Ada 83 program has use-visibility of any of these new identifiers, it is probable that the program is an illegal Ada 95 program, but it is possible (if the Ada 83 and Ada 95 uses of the identifiers are similar) that the program is a legal Ada 95 program with different semantics!  
For example:
package My_Characters is
	type Wide_Character is ('A', 'B', 'C', ...);
end My_Characters;
-----------------------------
with Text_Io;
with My_Characters;  use My_Characters;
procedure Example is
	Char : Wide_Character := 'C';
	Pos : Natural := Wide_Character'Pos (Char);
begin
	Text_Io.Put (Integer'Image(Pos));
end Example;
In Ada 83, the variable Char is of subtype My_Characters.Wide_Character and the string "2" will be output.  In Ada 95, the variable Char is of subtype Standard.Wide_Character and the string "67" will be output.
(f)	Avoidance in Ada 83
Treat all the identifiers in package Standard as reserved words.
(g)	Detection and Automated Correction
Detection of the incompatibility is straightforward.  Automated correction is possible, by replacing (for example) Wide_Character by Other_Package.Wide_Character, where Other_Package is the other use-visible package that exports the colliding identifier.
41.	New Pragmas Defined
(a)	Status  	Upward Inconsistent, Implementation-Dependent - no known problems
		Upward Consistent, Implementation-Dependent - no known problems
(b)	References  LRM-(various);  RM95-(various)
(c)	Language Change
New pragmas have been defined in Ada 95:
All_Calls_Remote (E.2.3), Asynchronous (E.4.1), Atomic (C.6), Atomic_Components (C.6), Attach_Handler (C.3.1), Convention (B.1), Discard_Names (C.5), Elaborate_All (10.2.1), Elaborate_Body (10.2.1), Export (B.1), Import (B.1), Inspection_Point (H.3.2), Interrupt_Handler (C.3.1), Interrupt_Priority (D.1), Linker_Options (B.1), Locking_Policy (D.3), Normalize_Scalars (H.1), Preelaborate (10.2.1), Pure (10.2.1), Queuing_Policy (D.4), Remote_Call_Interface (E.2.3), Remote_Types (E.2.2), Restrictions (13.12), Reviewable (H.3.1), Shared_Passive (E.2.1), Task_Dispatching_Policy (D.2.2), Volatile (C.6), Volatile_Components (C.6).
(d)	Rationale
New pragmas are needed to satisfy some of the requirements for Ada 95.
(e)	Incompatibility
Implementations are free to add pragmas to the language.  If an Ada 83 program uses an implementation-defined pragma with the same name, it is probable that the program is an illegal Ada 95 program (see Incompatibility 1), but it is possible (if the Ada 83 and Ada 95 uses of the pragmas are similar) that the program is a legal Ada 95 program with different semantics!
There are no known Ada 83 implementations that have implemented a pragma with the same name as one of the new pragmas, other than with the same semantics.
(f)	Avoidance in Ada 83
None possible, assuming use of the implementation-defined pragmas is required.  Perhaps implementations will provide an alternative name for the offending pragma long before Ada 95 compilers are available.
(g)	Detection and Automated Correction
Detection of the use of new Ada 95 pragmas is straightforward.  Manual correction is necessary as the semantics of implementation-defined pragmas are unavailable.
(h)	Notes
The same problem could arise when moving from one Ada 83 implementation to another.
42.	New Attributes Defined
(a)	Status  	Upward Inconsistent, Implementation-Dependent - no known problems
		Upward Consistent, Implementation-Dependent - no known problems
(b)	References  LRM-(various);  RM95-(various)
(c)	Language Change
New attributes have been defined in Ada 95:
Access (3.10.2), Adjacent (A.5.3), Alignment (13.3), Bit_Order (13.5.3), Body_Version (E.3), Caller (C.7.1), Ceiling (A.5.3), Class (3.9), Component_Size (13.3), Compose (A.5.3), Copy_Sign (A.5.3), Definite (12.5.1), Denorm (A.5.3), Exponent (A.5.3), External_Tag (13.3), Floor (A.5.3), Fraction (A.5.3), Identity (11.4.1), Input (13.13.2), Leading_Part (A.5.3), Machine (A.5.3), Max (3.5), Max_Size_In_Storage_Elements (13.11.1), Min (3.5), Model (A.5.3), Model_Emin (A.5.3), Model_Epsilon (A.5.3), Model_Mantissa (A.5.3), Model_Small (A.5.3), Modulus (3.5.4), Output (13.13.2), Partition_Id (E.1), Read (13.13.2), Remainder (A.5.3), Round (3.5.10), Rounding (A.5.3), Safe_First (A.5.3), Safe_Last (A.5.3), Scale (3.5.10), Scaling (A.5.3), Signed_Zeros (A.5.3), Storage_Pool (13.11), Tag (3.9), Truncation (A.5.3), Unbiased_Rounding (A.5.3), Unchecked_Access (13.10), Valid (13.9.2), Version (E.3), Wide_Image (3.5), Wide_Value (3.5), Wide_Width (3.5), Write (13.13.2).
(d)	Rationale
New attributes are needed to satisfy some of the requirements for Ada 95.
(e)	Incompatibility
Implementations are free to add attributes to the language.  If an Ada 83 program uses an implementation-defined attribute with the same name as an Ada 95 defined attribute, it is probable that the program is an illegal Ada 95 program, but it is possible (if the Ada 83 and Ada 95 uses of the attributes are similar) that the program is a legal Ada 95 program with different semantics!
There are no known Ada 83 implementations that have implemented an attribute with the same name as one of the new attributes.
(f)	Avoidance in Ada 83
None possible, assuming use of the implementation-defined attributes is required.  Perhaps implementations will provide an alternative name for the offending attribute long before Ada 95 compilers are available.
(g)	Detection and Automated Correction
Detection of the use of new Ada 95 attributes is straightforward.  Manual correction is necessary as the semantics of implementation-defined attributes are unavailable.
43.	New Library Units Defined
(a)	Status  	Upward Consistent, Implementation-Dependent - no known problems
(b)	References  RM95-A.2(2), B.2(3)
(c)	Language Change
New library units have been defined in Ada 95:
Ada (A.2), Interfaces (B.2).
(d)	Rationale
New library units are needed to satisfy some of the requirements for Ada 95.  The number of new library units required has been minimised by making all language-defined library units child units of packages Ada, Interfaces or System.
(e)	Incompatibility
Implementations and users are free to define new library units.  However, some implementations do not allow the language-defined library units to be redefined, so attempting to compile a user-defined library unit with the same name as a library unit added by Ada 95 could fail.  Any Ada 83 program using such a package would probably be an illegal Ada 95 program.
There are no known Ada 83 implementations that supply a package with one of the new names.
(f)	Avoidance in Ada 83
Avoid naming library units with the same names as Ada 95 defined packages.  Where the library unit is implementation-defined, no avoidance is possible, assuming use of the library unit is required.  Perhaps implementations will provide an alternative name for the offending library unit long before Ada 95 compilers are available.
(g)	Detection and Automated Correction
Detection of the definition and/or use of library units with the same name as new Ada 95 library units is straightforward.  For user-defined units, automatic correction is possible.  For implementation-defined names, manual correction is necessary.
Guidelines for Avoiding Transition Problems
This section describes a set of rules for writing Ada 83 programs to avoid upward incompatibilities when transitioning to Ada 95.  They are a simplified version of those described earlier in the guide and avoid all but the obscure incompatibilities.
(1)	Do not use the following identifiers:
(a)	Abstract, Aliased, Protected, Requeue, Tagged, Until (see Incompatibility 2);
(b)	Wide_Character, Wide_String (see Incompatibility 40);
(c)	Ada, Interfaces (see Incompatibility 43).
(2)	Do not apply a use-clause for package SystemAlternatively, avoid using the following identifiers: Any_Priority, Bit_Order, Default_Bit_Order, Default_Priority, Interrupt_Priority, Max_Base_Digits, Max_Binary_Modulus, Max_Nonbinary_Modulus, Null_Address, Word_Size  (see Incompatibility 36).
(3)	Do not apply a use-clause for package Text_Io nor for instantiations of package Sequential_IoAlternatively, avoid using the following identifiers: Current_Error, Decimal_Io, File_Access, Flush, Get_Immediate, Is_Open, Look_Ahead, Modular_Io, Set_Error, Standard_Error; (see Incompatibilities 37 and 39).
(4)	Do not provide a body for a library package that does not require oneAlternatively, place a pragma Elaborate_Body in the specification - this will be ignored in Ada 83 (unrecognised pragma) but will cause the body to be required (and hence be legal) in Ada 95 (see Incompatibility 28).
(5)	Add a distinctive commentsuch as -- ! (<>) in Ada 95 to all generic formal private types that can be legally instantiated with indefinite subtypesie unconstrained subtypes without defaults (see Incompatibility 31).
(6)	Place representation clauses for a real type immediately after the type declaration (see Incompatibility 7).
(7)	Do not derive from a type in the package declaring the parent type (see Incompatibility 4)Alternatively, derive any new types before redefining any predefined operations on the parent type..
(8)	Avoid use of the model-oriented attributes of real types (see Incompatibility 10).
(9)	Allow for subtype Character to have 256 positions (see Incompatibility 5) and for character literals to be always visible (see Incompatibility 25).
(10)	Avoid use of relational operators between character or string literals (see Incompatibilities 6 and 13).
(11)	Do not assume too much about the state of the computation when exceptions are implicitly raised.  Do not knowingly cause implicit exceptions to be raised (see Incompatibilities 8, 14, 15, 17, 26 and 38). 
(12)	Always handle Constraint_Error and Numeric_Error in the same exception handler (see Incompatibility 29).

